Life-Cycle Saving Problem

Yongyang Cai and Kenneth Judd

August 6, 2018

1 Life-Cycle Saving Problem

The life-cycle saving problem is to maximize the total utility

$$\sum_{t=1}^{T} \beta^t u(c_t)$$

subject to the constraints:

$$S_{t+1} = (1+r)S_t + w_{t+1} - c_{t+1}, \qquad 0 \le t < T$$

where S_t is the savings (assets), c_t is consumption, w_t is the wage rate at time t, and $r = 0.2$ is the interest rate, $\beta = 0.9$ is the discount factor and the concave utility function u over consumption c is

$$u(c) = -\exp(-c)$$

And we assume that assets are initially zero and terminally zero, i.e., $S_0 = S_T = 0$, and assume that $w_t = 1$ for $t < R$ and $w_t = 0$ for $t \ge R$, where R is the retiring year.

The following is the AMPL code:

```
param T := 100;                 # Number of periods
param R := 60;                   # Retirement
param beta := 0.9;              # discount rate
param r := 0.2;                    # interest rate
param S0 := 0;                   # initial savings
param ST := 0;                  # final savings
param w{1..T};      # wages
let {i in 1..(R-1)} w[i] := 1.0;
let {i in R..T} w[i] := 0.0;

var S{0..T};            # savings
var c{0..T};            # consumptions

maximize utility: sum {t in 1..T} beta^t*(-exp(-c[t]));
subject to budget {t in 0..T-1}: S[t+1] = (1+r)*S[t] + w[t+1] - c[t+1];
subject to savings {t in 0..T}: S[t] >= 0.0;
subject to consumptions {t in 1..T}: c[t] >= 0.0;
subject to bc1: S[0] = S0;
subject to bc2: S[T] = ST;
subject to bc3: c[0] = 0.0;
```